APPROVED BY
J. W. Snyder

IDENTIFICATION

Gauss Quadrature (Three-Point), MU GQF1

E. M. Zographos - January 11, 1957

Midwestern Universities Research Association, Madison, Wisconsin

PURPOSE

To evaluate $I = \int_{-1/2}^{+1/2} \phi(u)du$ in fixed point arithmetic.

RESTRICTIONS

No internal check is made on overflow, since it is left to the user to ensure that all numbers stay in the range $-(1 - 2^{-35})$ to $+(1 - 2^{-35})$.

METHOD

Procedure: The three-point Gauss quadrature formula

$$\int_{-1/2}^{+1/2} \phi(u)du \cong R_1\phi(u_1) + R_2\phi(u_2) + R_3\phi(u_3)$$

is used; where u_1, u_2, u_3 are the points of subdivision of the interval, $-1/2 \leq u \leq +1/2$, and R_1, R_2, R_3 are the weights.

The integral over the interval $(-1/2, +1/2)$ is related to the integral over the arbitrary interval (a, b) as follows:

$$\int_a^b f(x)dx = (b - a)\int_{-1/2}^{+1/2} \phi(u)du ,$$

where $\phi(u)$ is obtained from f(x) by making the change in variable

$$x = (b - a)\, u + \frac{a + b}{2} .$$

The $\phi(u)$'s are evaluated by a closed subroutine known as the AUXILIARY SUBROUTINE, which must be provided by the user. During each pass through MU GQF1 three entires are made into the AUXILIARY SUBROUTINE to obtain $\phi(u_1)$, $\phi(u_2)$, $\phi(u_3)$.

Accuracy: The three-point Gauss quadrature formula is exact when $\phi(u)$ is a polynomial of degree five or less. For $\phi(u)$ not a polynomial of degree five or less, the error is

$$\int_{-1/2}^{+1/2} \phi(u)du - \left[R_1\phi(u_1) + R_2\phi(u_2) + R_3\phi(u_3)\right] = 4.96 \times 10^{-7}\, \phi^{(6)}(\eta), \quad (-1/2 \leq \eta \leq +1/2).$$

The roundoff error is always less than 1.5×2^{-35} and therefore is negligible in comparison with the above truncation error (assuming $\phi^{(6)}(\eta)$ not vanishingly small).
(Cf. J. B. Scarborough, Numerical Mathematical Analysis, The John Hopkins Press, 1955, Chapter VII. Z. Kopal, Numerical Analysis, John Wiley and Sons, Inc., 1955, Chapter VII.)

USAGE

Calling Sequence:

UA SAP

Loc.	Instruction
α	TSX GQF1, 4
$\alpha + 1$	Return

MURASS

Loc.	Instruction
α	BTSX A 4
$\alpha + 1$	Return

Return is made with the value of the integral in the AC and in TGQF1 + 2.

The symbolic location of the AUXILIARY SUBROUTINE, WGQF1 (MURASS: W), must be defined by the user in the assembly of MU GQF1.

Space Required:

UA SAP		MURASS
21 words program at:	GQF1	A
3 words temporary at:	TGQF1	T

and space for the AUXILIARY SUBROUTINE.

The three words of temporary storage are not available for use by the AUXILIARY SUBROUTINE since they contain information which is to be used later by MU GQF1. However, when MU GQF1 does not have control, these three words of temporary storage are available.

Relocatable Cards: The nominal addresses in the relocatable program deck **have been** defined as follows:

SYMBOLIC ADDRESS	NOMINAL ADDRESS
GQF1	0
TGQF1	2000_8
WGQF1	3000_8

CODING INFORMATION

The AUXILIARY SUBROUTINE must be written as a closed subroutine: Entry **to the AUXILIARY SUBROUTINE** is made with u in the accumulator via

UA SAP

Loc.	Instruction
α	TSX WGQF1, 4
$\alpha + 1$	Return

MURASS

Loc.	Instruction
α	BTSX W 4
$\alpha + 1$	Return

and on exit Ø(u) must be in the accumulator.

Timing: [1.5 + 3 (AUXILIARY SUBROUTINE time in milliseconds)] milliseconds.

```
      REM  GQF1 MURA GAUSS QUADRATURE (THREE-POINT)                    MU   GQF1
GQF1  SXD  TGQF1,1          SAVE IR1                                   GQF1 0001
      SXD  TGQF1+1,4        SAVE IR4                                   GQF1 0002
      PXD  3,0              CLEAR AC                                   GQF1 0003
      STO  TGQF1+2          CLEAR SUM BOX                              GQF1 0004
      LXA  GQF1+2,1         SET IR1=3                                  GQF1 0005
      CLA  GQF1+18,1        U                                          GQF1 0006
      TSX  WGQF1,4          COMPUTE PHI (U)                            GQF1 0007
      LRS  35               PHI (U) TO MQ                              GQF1 0008
      MPR  GQF1+21,1        MULTIPLY BY R                              GQF1 0009
      ADD  TGQF1+2          ADD TO SUM                                 GQF1 0010
      STO  TGQF1+2                                                     GQF1 0011
      TIX  GQF1+5,1,1       LOOP RE-ENTRY, OUT AFTER 3 PASSES          GQF1 0012
      LXD  TGQF1,1          RESTORE IR1                                GQF1 0013
      LXD  TGQF1+1,4        RESTORE IR4                                GQF1 0014
      TRA  1,4              OUT                                        GQF1 0015
      OCT  543113757210     U SUB 1                                    GQF1 0016
      OCT  000000000000     U SUB 2                                    GQF1 0017
      OCT  143113757210     U SUB 3                                    GQF1 0018
      OCT  107070707071     R SUB 1                                    GQF1 0019
      OCT  161616161616     R SUB 2                                    GQF1 0020
      OCT  107070707071     R SUB 3                                    GQF1 0021
```